```
'--------------------------------------------------------------------------------------------+
'                                                                                            |
'                         Digital Transform Fourier Analysis                                 |
'                                                                                            |
'File: DigitalTransform.vbp                                                                  |
'Purpose:                                                                                    |
'     The program demonstrate the digial superposition of several harmonic signal.          |
'     And plot the corresponding Fourier transform spectrums.                                |
'     Users can add the harmonic function to the input stream. When program start,           |
'     it will perform sampling of the analogy input signal according the sampling            |
'     frequency and the number of sample of the input. After x[n] is obtained, the           |
'     further process is calculate the DFT X[k] of the input x[n]. Graphic in time           |
'     domain and frequnecy domain spectrums will also be ploted.                             |
'                                                                                            |
'Written by Dillian Wong, 8/4/04                                                             |
'--------------------------------------------------------------------------------------------+

Option Explicit

'_____________________ G L O B A L  V A R I A B L E S _____________________________________

Dim inputArray(2, 1000) As Double      'magnitude,function,period
Dim inputTerm As Integer               'Number of input term(eg 1+ cos2pi/4, 2 term)

Dim NoOfSample As Integer              'Number of samples that required
Dim samplingFrequency As Double        'Sampling Frequency of the analogy signal

Dim x(1000) As Double                  'input signal in time domain

Dim Fr(1000) As Double                 'Discrete Fourier Tranform in real part
Dim Fi(1000) As Double                 'Discrete Fourier Tranform in image part
Dim F(1000) As Double                  'Discrete Fourier Transform Magnitude
Dim Φ(1000) As Double


'_____________________ A C C E S S  I N P U T  P A R A M E T E R ___________________________
'/*
'  Function used to access the critical input variable "inputArray" and "InputTerm"
'*/
Public Sub setInputArray(a As Integer, b As Integer, d As Double)
    inputArray(a, b) = d
End Sub

Public Sub setInputTerm(a As Integer)
    inputTerm = a
End Sub


'/*
'  Add the harmonic function input signals, the inputArray will store the harmonic
'  signal data.
'*/
Private Sub cmdAdd_Click()

    'store the harmonic function entry into variable array
    inputArray(0, inputTerm) = txtMagnitude.Text

    If optFunction(0).Value = True Then
        inputArray(1, inputTerm) = 1             '1 means cos
    Else
        If optFunction(1).Value = True Then
            inputArray(1, inputTerm) = 0         '0 means sin
        Else
            inputArray(1, inputTerm) = -1        '-1 means DC
```

```
        End If
    End If
    inputArray(2, inputTerm) = txtPeriod.Text

    'point to next storage variable
    inputTerm = inputTerm + 1

    Call PrintEquation(inputArray, inputTerm)    'print out input signal
End Sub

'Form Loading
Private Sub Form_Load()
    LineXAxis.X1 = 0
    LineXAxis.X2 = picResponse.Width
    LineXAxis.Y1 = picResponse.Height / 2
    LineXAxis.Y2 = picResponse.Height / 2

    inputTerm = 0
End Sub


'_____________________ D F T  C A L C U L A T I O N _____________________________________________

'/*
'  Process the data, perform sampling and DTF. TextBox txtNoofSample and txtSamplingFrequency
'  are the input parmeter for sampling. variable inputArray is the oringinal harmonic signal
'  for sampling. After implementing the DFT transformation, print the output and input data
'  in files txtDFT.txt. Notice that this function will call program DFTMath.
'*/
Private Sub Calculate()
    Dim n As Integer
    Dim k As Integer
    Dim v As Variant

    'input parameter
    NoOfSample = txtNoOfSample.Text
    samplingFrequency = txtSamplingFrequency.Text

    'perform sampling of analogy signal
    v = Sampling(inputArray, inputTerm, NoOfSample, samplingFrequency)
    For n = 0 To 1000
        x(n) = v(n)
    Next n

    'perform DFT of x[n]
    v = DFT(x, NoOfSample)
    For k = 0 To 1000
        F(k) = v(k)
    Next k

   'store the data in file  "txtDFT.txt"
   Open App.Path & "\ txtDFT.txt" For Output As #1
        Print #1, "Equation : " & txtEquation.Text
        Print #1, "sampling Frequency: " & txtSamplingFrequency.Text
        Print #1, "No of sample : " & txtNoOfSample.Text
        Print #1, ""
        Print #1, ""
        'print input data out, time x[n]
        For n = 0 To NoOfSample - 1
        Print #1, "x[" & n & "]" & " = " & FormatNumber(x(n), 2)
        Next n

        Print #1, ""
        Print #1, ""
```

```
        For n = 0 To NoOfSample - 1
        Print #1, "F[" & n & "]" & " = " & FormatNumber(F(n), 2)
        Next n
   Close 1
End Sub



'_____________________ P L O T  G R A P H _____________________________________________________
'/*
'  Draw signal in Time/Frequency Domain. Before plot the spectrum, it will clear
'  the picture box first, then process the data once again to ensure any new data
'  input are inclued.
'*/
Private Sub cmdTimeDomain_Click()
    CallmnuClearResponse_Click
    Call Calculate
    Call printDataOut(x, F, NoOfSample, True)
    Call plot(x, NoOfSample, samplingFrequency)
End Sub
Private Sub cmdFrequencyDomain_Click()
    CallmnuClearResponse_Click
    Call Calculate
    Call printDataOut(x, F, NoOfSample, False)
    Call plot(F, 1000, samplingFrequency)      ' 1000 as DTF is repetative
End Sub



'_____________________ A B O U T _____________________________________________________
'// Description of the Program DFT V0.1
Private Sub mnuAbout_Click()
    MsgBox ("The program demonstrate the digial superposition of several harmonic signal." _
         & "And plot the corresponding Fourier transform spectrums." _
         & "Users can add the harmonic function to the input stream. When program start," _
         & "it will perform sampling of the analogy input signal according the sampling" _
         & "frequency and the number of sample of the input. After x[n] is obtained, the" _
         & "further process is calculate the DFT X[k] of the input x[n]. Graphic in time" _
         & "domain and frequnecy domain spectrums will also be ploted. ")
End Sub



'_____________________ M E N U  F U N C T I O N _____________________________________________

'// Clear the x[n]
Private Sub mnuClearResponse_Click()
    Dim i As Integer
    txtResponse.Text = ""
    picResponse.Cls
    For i = 0 To 1000
         x(i) = 0
         Fr(i) = 0
         Fi(i) = 0
         F(i) = 0
         Φ(i) = 0
    Next i
End Sub

'//Clear the equation x(t)
Private Sub mnuClearEquation_Click()
    Dim i, j, n As Integer
    For i = 0 To 1000
         x(i) = 0
```

```
         Fr(i) = 0
         Fi(i) = 0
         F(i) = 0
         Φ(i) = 0
    Next i
    For j = 0 To 2
        For i = 0 To 1000
            inputArray(j, i) = 0
        Next i
    Next j
    inputTerm = 0
    txtEquation.Text = ""
End Sub

'//Some sample for DTF demonstation
Private Sub mnuSample_Click(Index As Integer)
    CallmnuClearEquation_Click
    CallmnuClearResponse_Click
    Select Case Index
        Case 1: Call sample1
        Case 2: Call sample2
        Case 3: Call sample3
        Case 4: Call sample4
    End Select

    Call PrintEquation(inputArray, inputTerm)
    Call Calculate
End Sub
```

```
'-------------------------------------------------------------------------------------------+
'                                                                                           |
' File: DFT.bas                                                                            |
' Purpose:                                                                                 |
'        Perform Discrete Fourier Transformation. When oringinal harmonic signal is sampled. |
'        We have x[n] and input parameter that the number of sample of one period N, thus   |
'        sufficient infomation to implement DTF. The program is divided into two parts, the |
'        first perform DTF and get the N samples F[k], the second part copy the N samples   |
'        along the frequency axis. The characteristic is because of the sampling.           |
'                                                                                           |
'-------------------------------------------------------------------------------------------+

'_____________________ M A T H  C A L C U L A T I O N _____________________________________

Public Function DFT(x() As Double, NoOfSample As Integer) As Variant

    Dim n As Integer
    Dim ans As Single              'Temporary answer
    Dim T As Double
    Dim pi As Double               '3.14..
    Dim i As Integer
    Dim k As Integer

    Dim Fr(1000) As Double         'real part of F
    Dim Fi(1000) As Double         'image part of F
    Dim F(1000) As Double          'magnitude of F
    Dim Φ(1000) As Double          'phase of F


    pi = 4 * Atn(1)

   '******Perform DTF transformation******
   For k = 0 To NoOfSample - 1

        'Real Part
        ans = 0
        For n = 0 To NoOfSample - 1
            ans = ans + x(n) * Math.Cos(2 * pi * k * n / NoOfSample)
        Next n
        Fr(k) = ans

        'Image Part
        ans = 0
        For n = 0 To NoOfSample - 1
            ans = ans + x(n) * Math.Sin(2 * pi * k * n / NoOfSample)
        Next n
        Fi(k) = ans

      'magnitude
      F(k) = (Fr(k) ^ 2 + Fi(k) ^ 2) ^ 0.5
      If Fi(k) = 0 Then
        Φ(k) = pi / 2
      Else
        Φ(k) = Math.Atn(Fr(k) / Fi(k))  'inverse trigonometric function of Tan
      End If

   Next k


   '******Fill the repetitive samples in Frequency Domain******
    'fill N samples signal in the time domain and form a peridic signal
    i = 0
    For k = NoOfSample To 1000
       F(k) = F(i)
```

```
       i = i + 1
       If i = NoOfSample Then i = 0
    Next k

    DFT = F()       '* return DFT calculation result to caller

End Function
```

Draw - 1

```
'-------------------------------------------------------------------------------------------+
'                                                                                           |
' File: Draw.bas                                                                           |
' Purpose:                                                                                 |
'        This function will plot the spectrum in time domain or in frequency domain. Pass  |
'        the data x[n] or F[n], Number of samples and sampling Freqeuncy to function, will |
'        plot the data in the picture box. the magnitude will fix the scalar of the picture|
'        box, and the step in x-axis will rearragned according the sampling frequency. All |
'        in all, magnutide and the step in x-asix is necessary to fix the picture box.     |
'                                                                                           |
'-------------------------------------------------------------------------------------------+

'_____________________ M A T H  C A L C U L A T I O N _____________________________________

Public Sub plot(x() As Double, NoOfSample As Integer, samplingFrequency As Double)

    Dim n As Integer
    Dim magnitude As Integer
    Dim xAxis As Long
    Dim xstep As Integer
    Dim max As Integer

    frmMain.picResponse.Cls
    frmMain.picResponse.DrawWidth = 3

    'step of x-axis
    xstep = FormatNumber(1 / samplingFrequency * 500, 6)
    xAxis = 720

    'weight of manitude, determine the max-x[n] so that to match the height of picResponse
    max = 0
    For n = 0 To NoOfSample - 1
        If Abs(x(n)) > max Then
            max = Abs(x(n))
        End If
    Next n


    'draw line
    For n = 0 To NoOfSample - 1
       xAxis = xAxis + xstep

        If xAxis < frmMain.picResponse.ScaleWidth Then
            'max magnitude only consume 80% of the max height of the picture box
            magnitude = x(n) * (frmMain.picResponse.ScaleHeight / 2) * 0.8 / max
            frmMain.picResponse.Line (xAxis, frmMain.picResponse.ScaleHeight / 2)-(xAxis, frmMain.picResponse.Scal
eHeight / 2 - magnitude)
        End If
    Next n

End Sub
```

```
printOut - 1

'*******************************************************************************
'/*
'   print the harmonic sampled signal x[n] and DTF transfromed F[k] in the text
'   box.
'*/
'*******************************************************************************
Public Sub printDataOut(x() As Double, F() As Double, NoOfSample As Integer, isT As Boolean)

    Dim k As Integer
    Dim strxn As String

    frmMain.txtResponse.Text = ""


    If isT = True Then
        'print input data out, time x[n]
        strxn = ""
        For n = 0 To NoOfSample - 1
            strxn = strxn & FormatNumber(x(n), 2) & " , "
        Next n
            strxn = strxn & "..... + repeat term"
        frmMain.txtResponse.Text = frmMain.txtResponse.Text & " x[n] = { " & strxn & " } "
    End If


    If isT = False Then
        'print input data out, frequency F[k]
        strxn = ""
        For k = 0 To NoOfSample - 1
            strxn = strxn & FormatNumber(F(k), 2) & " , "
        Next k
            strxn = strxn & "..... + repeat term"
        frmMain.txtResponse.Text = frmMain.txtResponse.Text & " F[k] = { " & strxn & " } "
    End If

End Sub



'/*
'    Print harmonic function input signals, in the label. So that user can know
'    what is his/her input signals
'*/
Public Sub PrintEquation(inputArray() As Double, inputTerm As Integer)

    Dim n As Integer
    Dim s As String

    'write down the user input harmonic function
    frmMain.txtEquation.Text = ""
    For n = 0 To inputTerm - 1

        'determine whether it is COS or SIN function
        If inputArray(1, n) = 1 Then
            s = "COS(2πn/"
        Else
            If inputArray(1, n) = 0 Then
                s = "SIN(2πn/"
            Else
                s = ""
            End If
        End If

        'print out
```

```
printOut - 2

        If Not frmMain.txtEquation.Text = "" Then
            frmMain.txtEquation.Text = frmMain.txtEquation.Text & " + "
        End If
        If Not inputArray(1, n) = -1 Then
            frmMain.txtEquation.Text = frmMain.txtEquation.Text & inputArray(0, n) & s & inputArray(2, n) & ")"
        Else
            frmMain.txtEquation.Text = frmMain.txtEquation.Text & inputArray(0, n)
        End If

    Next n

End Sub
```

Sample - 1

```
'*****************************************************************************
'/*
'   Following is the samples for the DFT analysis
'*/
'*****************************************************************************

'/*
'  EXAMPLE 1:
'  x(t) = 4
'  DC with magnitude 4
'*/
Public Sub sample1()
    'inputArray(field1,field2)
    '[field1]
    'magnitude,function,period
    '--magnitude:
    '--function: 1 means cos, 0 means sin, -1 means DC
    '--period: sec, for DC period is meaningless
    '[field2]
    'term index
    Call frmMain.setInputArray(0, 0, 4)
    Call frmMain.setInputArray(1, 0, -1)
    Call frmMain.setInputArray(2, 0, 10)

    CallfrmMain.setInputTerm(1)

    'set text field display
    frmMain.txtNoOfSample.Text = 20
    frmMain.txtSamplingFrequency.Text = 5

End Sub



'/*
'  EXAMPLE 2:
'  x(t) = 1 + 2COS(2πn/4) + 1SIN(2πn/8)
'  sampling frequency 5 and no of sample = 10
'*/
Public Sub sample2()
    Dim i As Integer
    i = 0

    Call frmMain.setInputArray(0, 0, 1)
    Call frmMain.setInputArray(1, 0, -1)
    Call frmMain.setInputArray(2, 0, 1)
    i = i + 1

    Call frmMain.setInputArray(0, 1, 2)
    Call frmMain.setInputArray(1, 1, 1)
    Call frmMain.setInputArray(2, 1, 4)
    i = i + 1

    Call frmMain.setInputArray(0, 2, 1)
    Call frmMain.setInputArray(1, 2, 0)
    Call frmMain.setInputArray(2, 2, 8)
    i = i + 1

    CallfrmMain.setInputTerm(i)

    frmMain.txtNoOfSample.Text = 10
    frmMain.txtSamplingFrequency.Text = 5

End Sub
```

Sample - 2

```
'/*
'  EXAMPLE 3:
'  x(t) = cos(2π n/4)
'  fs = 10, N=100
'*/
Public Sub sample3()
    Dim i As Integer
    i = 0

    Call frmMain.setInputArray(0, 0, 2)
    Call frmMain.setInputArray(1, 0, 1)
    Call frmMain.setInputArray(2, 0, 4)
    i = i + 1

    CallfrmMain.setInputTerm(i)

    frmMain.txtNoOfSample.Text = 100
    frmMain.txtSamplingFrequency.Text = 10

End Sub

'/*
'  EXAMPLE 4:
'  x(t) = cos(2π n/4)+ cos(2π n/8)+cos(2π n/12)
'  fs = 25, N = 600
'*/
Public Sub sample4()
    Dim i As Integer
    i = 0

    Call frmMain.setInputArray(0, 0, 2)
    Call frmMain.setInputArray(1, 0, 1)
    Call frmMain.setInputArray(2, 0, 4)
    i = i + 1

    Call frmMain.setInputArray(0, 1, 2)
    Call frmMain.setInputArray(1, 1, 1)
    Call frmMain.setInputArray(2, 1, 8)
    i = i + 1

    Call frmMain.setInputArray(0, 2, 2)
    Call frmMain.setInputArray(1, 2, 1)
    Call frmMain.setInputArray(2, 2, 12)
    i = i + 1

    CallfrmMain.setInputTerm(i)

    frmMain.txtNoOfSample.Text = 600
    frmMain.txtSamplingFrequency.Text = 25

End Sub
```

```
'*********************************************************************************************
'/*
'   Perform sampling of the analogy harmonic signal. inputArray consist the harmonic function
'   parameter, like the magnitude, sin/Cos/Dc and period. inputTerm indicate how many input
'   term of harmonic function. Of course, No of sample and sampling frequecy are the parameter
'   for sampling.
'*/
'*********************************************************************************************
Public Function Sampling(inputArray() As Double, inputTerm As Integer, NoOfSample _
                         As Integer, samplingFrequency As Double) As Variant

    Dim n As Integer
    Dim ans As Double
    Dim T As Double
    Dim pi As Double
    Dim i As Integer
    Dim k As Integer
    Dim x(1000) As Double

    pi = 4 * Atn(1)


    '*Process data in Time Domain
    For n = 0 To NoOfSample - 1
        For i = 0 To inputTerm - 1

            If inputArray(1, i) = 1 Then        ' cos signal
                ans = inputArray(0, i) * Math.Cos(2 * pi * n / (inputArray(2, i) * samplingFrequency))
            Else
                If inputArray(1, i) = 0 Then    ' sin signal
                    ans = inputArray(0, i) * Math.Sin(2 * pi * n / (inputArray(2, i) * samplingFrequency))
                Else                            ' DC signal
                    ans = inputArray(0, i)
                End If
            End If
            x(n) = x(n) + ans
        Next i

      'print out
      'frmMain.lstResponse.AddItem (" x[" & n & "]  = " & FormatNumber(x(n), 6))
   Next n

   Sampling = x()

End Function
```